

THE IRISH NATURALISTS' JOURNAL

Vol. 22

APRIL 1986

No. 2

A COMPUTER PROGRAMME FOR CONVERSION OF THE IRISH TRANSVERSE MERCATOR PROJECTION TO THE UNIVERSAL TRANSVERSE MERCATOR PROJECTION

PIERRE RASMONT

*Zoologie Générale et Appliquée, Faculté des Sciences Agronomiques de l'État,
Gembloux, Belgium*

MARTIN C. D. SPEIGHT

Research Branch, Forest and Wildlife Service, Bray, Co Wicklow

BERNARD E. PICTON

Ulster Museum, Botanic Gardens, Belfast

At a recent meeting of the European Invertebrates Survey Committee (EIS), the feasibility of producing a computer programme for converting Irish Grid references to Universal Transverse Mercator (UTM) references was discussed. This conversion involves complex mathematics because the two grid systems have different geographical points of origin. Subsequently one of us (PR) has produced a computer programme which makes this conversion. The programme is reproduced as the final section of this text. The introductory notes and an optional input and output routine for latitude and longitude (calculated in the original programme as an intermediate step in the conversion) have been added by MS and BP.

In ecological work in general and nature conservation work in particular the importance of having accurate up-to-date information on the distribution of animal and plant species is well recognised. The interpretation of local distribution data is to a significant extent dependent upon national distribution data and a similar relationship exists between national and international data. In order for distribution data from different parts of Europe to be either combined or compared effectively some system of uniform presentation of data is required. Because the UTM grid provides such a uniform system, worldwide, it has been adopted by Europe's biologists for plotting international distribution data. For example, the maps in published parts of the *Atlas Florae Europaeae* are all on the UTM grid. Increasingly, national distribution data are being plotted on UTM maps as well, for instance in Belgium, all the Scandinavian countries, Austria, Switzerland, etc.

Storage, analysis and retrieval of biological records data have until recently been enormously time-consuming. The advent of computers has revolutionised that situation. The rapid advances in computer technology occurring today have brought within reach of the average household micro-computers that have the capacity of the earlier mainframe computers. It is clear that the brief monopoly held by large institutions on the computer handling of biological records data is already ending. In the none-too-distant future biological recorders are as likely to hold their data on magnetic disc as are national records centres. Unfortunately as yet few specialist programmes are available to suit the needs of biological recorders.

The UTM grid and its use in mapping the distribution of organisms

The Universal Transverse Mercator system is a system of map projection designed for use between 80°S and 84°N all round the world. It originated in the U.S.A. and is also used in Canada, Mexico and most of Europe. The maps are based on meridians at 6° longitude intervals. The longitudinal zones are numbered 1 to 60 starting from 180° in an easterly direction. The latitudinal zones are lettered from 80°S starting with letter C and omitting I




Figure 1. Ireland, showing lettering of UTM 100km squares and numbering convention for 50km sub-squares.




Figure 2. Europe; showing numbering of UTM longitude zones and lettering of UTM latitude zones. N.B. By convention the west coast of Norway is included in longitudinal zone 32 and not in 31 as shown.

and O, there are 20 latitudinal zones. Within these zones 100km intervals have letter codes. A different origin for the lettering of 100km northing intervals in adjoining fuses is used to avoid confusion, odd numbered segments are lettered from A to V (20 letters omitting O and I) starting with A at the Equator, even numbered segments starting with F at the Equator and using the same sequence. Longitudinal lettering is based on the sequence A to Z, again omitting O and I, starting at 180°. This gives 24 letters, in groups of 8 for each 6° longitudinal segment (see figs 1 and 2).

The vast majority of distribution maps produced in Britain and Ireland have used the 10km square unit. The unit used for plotting the international distribution of organisms is the 50km UTM square, and an internationally adopted system of classification of these squares is used. The 100km square is divided into four sub-squares and these are numbered 1 to 4 as shown in fig 2. Each 50km square therefore has a full reference in the form 30 U PV.2 which can be abbreviated to PV.2 without repetition in Ireland. Note that some recording schemes reduce the horizontal dimensions of some squares when they fall near a line of compensation (eg. 6°W) and add small portions to some other squares. On land in Ireland this only applies to sectors PA and PB.

A conventional system of numbering a small square within a 100km square is used, the bottom left corner is the origin eg PV000000 and a 100m square 33km east and 55km north of this has the reference PV330550.

An author may consider that the distribution data he or she has available for a species is inadequate to justify mapping its distribution but is nonetheless worthy of publication. In such an instance the addition of a UTM reference will make the data useable internationally. This approach has been adopted in recent articles on Irish Otitidae (Speight *et al.* 1983) and Psilidae (Speight *et al.* 1985). With a conversion programme now available such UTM references could in future be given to the same accuracy as the original Irish reference. These two articles also include examples of UTM maps showing species' distributions in Ireland, using the 50km unit.

The role of the conversion programme

The programme provides a means whereby a distribution record for anywhere in Ireland, given on the Irish national grid, can be instantly converted into its UTM equivalent. This enables any individual or institution in Ireland to contribute directly to international

distribution schemes. Equally, it provides the means for any interested party to map the Irish distribution of any organism on the UTM format. We are not trying to suggest that mapping of distribution records is all there is to biological records activity; habitat recording, collecting of ecological information, etc. are equally important, as has been discussed elsewhere (see Speight 1977).

In Ireland use of four figure references (1km squares) is the norm. Biologists preferring the six figure notation (100m squares) commonly used in England (where 1:50000 scale maps are more easily available) and using the conversion programme presented here, can enter six figure references and obtain a six figure output. The programme is accurate to about 10m and is therefore capable of producing correct eight figure references. More accurate references can be entered if desired by an alternative entry routine. This produces a numerical UTM output to the nearest metre.

Marine scientists usually use latitude and longitude for positions to define the location of sampling sites and may find the option of lat/long output and input useful. The conversion is accurate to 10m, better than most position fixing systems. In addition the lat/long input can be used to produce a UTM output for anywhere in western Europe.

Individuals with a computer using a different dialect of Basic to the one used here (Microsoft Basic) will need to convert the programme to run on their machine, but this should not be difficult as Microsoft Basic is one of the easier dialects to adapt.

```

10 REM IRISH GRID REFERENCE TO UTM REF. PROGRAMME
20 REM FORMATTED FOR 80 COLUMN SCREEN
30 REM MICROSOFT BASIC-86 Rev. 5.22
40 DEFDBL B,E,F,G,H,K,L,M,N,P,Q,R,X,Z
50 DIM K(6),L(5),M(8),S(1),T(2),X(15),B$(2),Y$(2),Z$(2)
60 PI=3.1415926536E :K=PI/180
70 REM CLEAR SCREEN FUNCTION
80 CL$=CHR$(27)+"*": DOWN$=CHR$(10)+CHR$(10) : BACK$=CHR$(8)+CHR$(8)+CHR$(8)
90 FIN$ = "When finished type END" + CHR$(10)
100 REM INPUT OF DATA CONSTANTS
110 DATA 1.005034396, 5.044926593D-3, 1.055158829D-5, 2.059762235D-8
120 DATA 0.9983172081, 2.525251575D-3, 2.661454803D-6, 3.461562818D-9
130 DATA 6.670534610748291D-3, 6.768170197224D-3, 1.000035, 6377563.396
140 DATA 6.722670023D-3, 6378388, 500000, 0.9996
150 FOR I=0 TO 15:READ X(I):NEXT I:PRINT CL$+DOWN$
160 REM USER INSTRUCTIONS
170 PRINT TAB(14)"PROGRAMME FOR CONVERSION OF IRISH GRID REFERENCES": PRINT
180 PRINT TAB(17)"TO UNIVERSAL TRANSVERSE MERCATOR REFERENCES": PRINT:PRINT
190 PRINT"Do you want to enter Irish Grid references (0) or Lat/Long (1) [0] "
    +BACK$;
200 AN$=INPUT$(1) :IF AN$ = "1" GOTO 1270
210 AN$="":PRINT:PRINT "If you want Lat/Long output enter Y [N] "+BACK$;
220 AN$ = INPUT$(1) : PRINT CL$+DOWN$
230 PRINT "Enter Irish Grid Reference - 4 or 6 figures":PRINT
240 PRINT "e.g. D1241 or D123413 or EXACT ": PRINT :PRINT FIN$
250 INPUT "IRISH REFERENCE = ",REF$
260 IF REF$ = "END" OR REF$ = "end" THEN GOTO 1100
270 IF LEN(REF$) < 3 THEN PRINT CL$+DOWN$ :GOTO 230
280 IF ASC(MID$(REF$,2,1)) <48 OR ASC(MID$(REF$,2,1)) >57 THEN REF$ = "EXACT":
    GOSUB 1180 :GOTO 320
290 REM SPLITS REFERENCE INTO LETTER, EASTING AND NORTHING; VALIDATES INPUT
300 IF LEN(REF$) =5 THEN B$(0)=LEFT$(REF$,1):B$(1)=MID$(REF$,2,2):B$(2)=RIGHT$(
    REF$,2):GOTO 320
310 IF LEN(REF$)= 7 THEN B$(0)=LEFT$(REF$,1):B$(1)=MID$(REF$,2,3):B$(2)=RIGHT$(
    REF$,3) ELSE PRINT "Wrong number of figures in reference":PRINT:GOTO 230
320 IF ASC(B$(0))>96 AND ASC(B$(0)) <123 THEN B$(0)=CHR$(ASC(B$(0))-32)
330 REM ASSIGNS NUMERICAL VALUE TO GRID SQUARE LETTER
340 G=0:H=0
350 H=INSTR("VWXYZQRSTULMNOPFGHJKABCDE",B$(0))
360 G=INSTR("VQLFAWRMGBXSNHCYTOJDZUPKE",B$(0))
370 IF G=0 OR H=0 THEN PRINT "There is no Irish grid square ";B$(0):PRINT:GOTO
    230
380 PRINT CL$ + DOWN$:PRINT "IRISH REFERENCE ";B$(0);B$(1);B$(2): PRINT DOWN$
390 G=(G-1)\5:H=(H-1)\5
400 REM CONVERTS IRISH T.M. COORDINATES TO METRES
410 IF LEN (B$(1))=3 AND REF$ <> "EXACT" THEN C=VAL(B$(1))/10:D=VAL(B$(2))/10:
    GOTO 440
420 C=VAL(B$(1)):D=VAL(B$(2))
430 IF C<0 OR C>100 OR D<0 OR D> 100 GOTO 230
440 X=1000!*(100!*G+C):Y=1000!*(100!*H+D):N=Y+5680030.5E

```

```

450 REM CONVERTS IRISH T.M. COORDINATES TO DEG/GREENWICH
460 FOR I=1 TO 6:K(I)=X(10)^I:NEXT I
470 M(0)=N/(X(11)*(1-X(8))*X(10)):Z=M(0)/X(0):G=0
480 WHILE M(2)>=.001 OR G<5
490 M(1)=X(0)*Z-X(1)/2*SIN(2*Z)+X(2)/4*SIN(4*Z)-X(3)/6*SIN(6*Z)
500 M(2)=M(0)-M(1):M(3)=M(2)/X(0):M(2)=ABS(M(2)*X(11)*(1-X(8))*X(10))
510 Z=Z+M(3): G=G+1
520 WEND
530 S(0)=SIN(Z):S(1)=S(0)^2:F=COS(Z):G=F^2:H=F^4:T(0)=TAN(Z)
540 T(1)=T(0)^2:T(2)=T(0)^4:M(4)=X(11)/SQR(1-X(8)*S(1))
550 M(5)=T(0)/(2*M(4)^2*K(2))*(1+X(8)*G)*100000000000E
560 M(6)=5+3*T(1)+6*X(9)*G-6*X(9)*S(1)-3*X(9)^2*H-9*X(9)^2*G*S(1)
570 M(6)=T(1)/(24*M(4)^4*K(4))*1D+24*M(6)
580 M(7)=1000000/(M(4)*X(10)*F)
590 M(8)=1/(F*6*(M(4)^3)*K(3))*(1+2*T(1)+X(9)*G)*1D+18
600 IF X>200000! THEN K(0)=X-200000! ELSE K(0)=200000!-X
610 Q=K(0)*.000001:L(0)=45*X(9)*T(1)*S(1): R=(M(4)/10000!)^6
620 L(0)=61+90*T(1)+45*T(1)+45*T(2)+107*X(9)*G-162*X(9)*S(1)-L(0)
630 L(0)=Q^6/(720*R*K(6))*T(0)*L(0)*1000000000000E
640 L(1)=1/(F*120*M(4)^5*K(5))*(5+28*T(1)+24*T(2)+6*X(9)*G+8*X(9)*S(1))
650 L(1)=Q^5*L(1)*1D+30
660 L(5)=Z-M(5)*Q^2+M(6)*Q^4-L(0):L(3)=M(7)*Q-M(8)*Q^3+L(1): D = 352
670 IF X>200000! THEN L(4)=D+L(3)/K ELSE L(4)=D-L(3)/K
680 LAT = L(5)/K :LONG = L(4)
690 REM CONVERTS LAT/LONG FROM DEGREES TO DEGREES AND DECIMAL MINUTES
700 IF ANS$ <> "Y" THEN GOTO 770
710 EW$="E"
720 IF LONG > 180 THEN LONG = 360 - LONG : EW$="W"
730 MINT = 100+ INT((LAT -INT(LAT))*6000+.5)/100: MINT$ = MID$(STR$(MINT),3,5)
740 MING = 100+INT((LONG-INT(LONG))*6000+.5)/100: MING$ = MID$(STR$(MING),3,5)
750 LONG$ = RIGHT$(STR$(INT(LONG)+1000),3) : LAT$ = STR$(INT(LAT))
760 PRINT "LATITUDE ";LAT$;" ";MINT$;" "N LONGITUDE ";LONG$;" ";MING$;" ";EW$
: PRINT
770 REM DETERMINATION OF THE U.T.M. ZONES
780 T(1)=INT(L(4))\6+1 : T(2)= (T(1)+29) MOD 60 +1: C=T(1)*6
790 IF C>180 THEN C=C-360
800 D=C-3 : Z$(0) = STR$(T(2)) :U=INT(LAT)\8+1 :U$=MID$("NPQRSTUVWXYZ",U,1)
810 REM DEG/GREENWICH TO NUMERIC U.T.M. COORDINATES
820 L(2)=L(4)*K: E=K*D: F=SIN(L(5)): G=COS(L(5)): H=(F/G)^2
830 K(1)=X(12)*G^2/(1-X(12)):K(2)=X(13)/SQR(1-X(12)*F^2):L=L(2)-E
840 IF ABS(L) <= PI THEN 860
850 L=L-2*PI*SGN(L)
860 M(1)=L*G:M(2)=M(1)^2:N=(13-64*H+K(1)*(4-24*H))*K(1)+14-58*H
870 M(3)=M(1)*(1+M(2)/6*(1-H+K(1)+M(2)/20*(5-H*(18-H)+K(1)*N))
880 M(3)=X(14)+K(2)*X(15)*M(3)
890 M(4)=X(4)*L(5)-X(5)*SIN(2*L(5))+X(6)*SIN(4*L(5))-X(7)*SIN(6*L(5))
900 M(6)=1+M(2)/12*(5-H+K(1)*(9+4*K(1)))
910 M(6)=X(15)*(X(13)*M(4)+K(2)*.5*L*M(1)*F*M(6))
920 Z=INT(M(3)*.01):P=INT(M(6)*.01):E=(INT(M(3)))/1000:N=(INT(M(6)))/1000
930 REM NUMERIC TO ALPHANUMERIC U.T.M. COORDINATES
940 J=INT(P/1000) : J=J MOD 20 +1 : T=INT(Z/1000)
950 IF INT(T(2))/2 = INT(T(2)/2) GOTO 970
960 Y$(1)=MID$("ABCDEFGHIJKLMNPQRSTUVWXYZ",J,1) : GOTO 980
970 Y$(1)=MID$("FGHIJKLMNPQRSTUVWXYZ",J,1)
980 ON T(2) MOD 3 + 1 GOSUB 990,1000,1010 :GOTO 1020
990 Z$(1)=MID$("STUVWXYZ",T,1) : RETURN
1000 Z$(1)=MID$("ABCDEFGH",T,1) : RETURN
1010 Z$(1)=MID$("JKLMNPQR",T,1) : RETURN
1020 Y$(2)=RIGHT$(STR$(P),3) : Z$(2)=RIGHT$(STR$(Z),3)
1030 IF REF$ = "EXACT" THEN REF$ = "*EXACT*"
1040 GOSUB 1110
1050 IF LEN(REF$)=5 THEN Y$(2)=LEFT$(Y$(2),2) : Z$(2)=LEFT$(Z$(2),2)
1060 PRINT "UTM REFERENCE IS "Z$(0);" ";U$;" ";Z$(1);Y$(1);" ";Z$(2);Y$(2)
: PRINT
1070 IF REF$ = "*EXACT*" THEN PRINT "EASTING ";E;"km NORTHING ";N;"km":PRINT
1080 REM RETURNS TO START FOR NEXT INPUT
1090 PRINT DOWN$ : IF AN$="1" GOTO 1280 ELSE GOTO 230
1100 PRINT CL$ : END
1110 REM OPTIONAL SUBROUTINE : PRINTS OUT 50 KM GRID SQUARE CODE
1120 IF VAL(Z$(2)) <500 AND VAL(Y$(2))=>500 THEN W$="1"
1130 IF VAL(Z$(2))=>500 AND VAL(Y$(2))=>500 THEN W$="3"
1140 IF VAL(Z$(2)) <500 AND VAL(Y$(2)) <500 THEN W$="2"
1150 IF VAL(Z$(2))=>500 AND VAL(Y$(2)) <500 THEN W$="4"
1160 PRINT "THE 50 KM SQUARE IS ";Z$(1);Y$(1);".";W$: PRINT :RETURN
1170 REM ENTRY SUBROUTINE FOR 0.0001 km PRECISION INPUT
1180 PRINT CL$+DOWN$:PRINT "ENTER IRISH 100km SQUARE LETTER eg D : ";
1190 B$(0)=INPUT$(1) :PRINT B$(0)

```

```

1200 PRINT "ENTER IRISH EASTING (km) eg. 11.321 : ";
1210 INPUT "", B$(1)
1220 PRINT "ENTER IRISH NORTHING (km) eg. 46.215 : ";
1230 INPUT "", B$(2)
1240 B$(1) = " " + B$(1) : B$(2) = " " + B$(2)
1250 RETURN
1260 REM INPUT ROUTINE FOR LAT/LONG
1270 PRINT CL$+DOWN$
1280 PRINT FIN$ : PRINT "ENTER LATITUDE degrees N : ";
1290 INPUT "", N$
1300 IF N$ = "END" OR N$ = "end" THEN GOTO 1100
1310 PRINT "ENTER mins eg. 15.01 : ";
1320 INPUT "", MINN$
1330 PRINT "IS LONGITUDE East or West : [W] "+BACK$;
1340 EW$=INPUT$(1) : IF EW$ <> "E" THEN EW$="W"
1350 PRINT CHR$(13) "ENTER LONGITUDE degrees ";EW$; " : "+BACK$;
1360 INPUT "", W$
1370 PRINT "ENTER mins eg. 12.23 : ";
1380 INPUT "", MINW$
1390 LAT = 10/600*VAL(MINN$) : LAT = LAT + VAL(N$)
1400 LONG = 10/600 *VAL(MINW$) : LONG = LONG + VAL(W$)
1410 IF EW$="W" THEN L(4) = 360-LONG ELSE L(4) = LONG
1420 NOTE$ = CL$ + DOWN$ + "LAT/LONG out of range " + DOWN$
1430 IF LAT >=80 OR LAT <0 OR LONG>180 OR LONG<0 THEN PRINT NOTE$ : GOTO 1280
1440 L(5) = LAT*K : PRINT CL$+DOWN$ :GOTO 730

```

NOTES ON THE MODULES AND VARIABLES USED IN THE PROGRAMME

Input of Irish Coordinates.

Input: Irish six figure or four figure reference; facility for input of exact Irish reference to 0.1 metre precision.

Output: B\$(0) = Letter of Irish T.M. square
 C = Irish T.M. easting (km) in the 100km square.
 D = Irish T.M. northing (km) in the 100 km square.

Alphanumeric to numeric Irish T.M. coordinates.

Output: X = numeric Irish T.M. easting (m).
 Y = numeric Irish T.M. northing (m).
 N = Irish T.M. northing translated to the Equator.

Numeric Irish T.M. coordinate to Degrees (East of Greenwich).

Input: X(0), X(1), X(2), X(3) = parameters of the Irish T.M. projection, calculated from eccentricity.
 X(8) = eccentricity (e^2 of the Airy ellipsoid).
 X(9) = arbitrary parameter.
 X(10) = scale coefficient of the Irish T.M. projection.
 X(11) = major axis of the Airy ellipsoid.

Output: LAT = latitude (deg).
 LONG = longitude (deg).
 L(5) = latitude (radians).
 L(4) = longitude (radians).

Determination of the U.T.M. zone.

Output: T(2) = U.T.M. zone number.
 D = central meridian of the U.T.M. fuse.
 U = Latitudinal zone.
 U\$ = Latitudinal code letter.

Degrees of Latitude and Longitude to numeric U.T.M. coordinates.

Input: X(4), X(5), X(6), X(7) = parameters of the U.T.M. projection calculated from eccentricity.
 X(12) = eccentricity (e^2 of the Hayford ellipsoid).
 X(13) = half large axis of the Hayford ellipsoid.
 X(14) = 500 000m (X-axis coordinate of the origin).
 X(15) = scale coefficient of the U.T.M. projection.

Output: Z = U.T.M. easting (0.1km) (integer).
 P = U.T.M. northing (0.1km) (integer).
 E = U.T.M. easting (km) (1m precision).
 N = U.T.M. northing (km) (1m precision).

Numeric to Alphanumeric U.T.M. coordinates.

Output: Z\$(1) = easting letter of the U.T.M. square.
 Y\$(1) = northing letter of the U.T.M. square.
 Z\$(2) = easting in the 100km U.T.M. square.
 Y\$(2) = northing in the 100km U.T.M. square.

Printing of Output.

Output matched to precision of input; four figure Irish reference produces four figure (1km) U.T.M. output; six figure Irish input produces six figure (0.1km) U.T.M. output; use of optional full precision Irish input routine called by entering EXTRA as Irish reference produces six figure U.T.M. reference and full numeric U.T.M. easting and northing.

Additional option for printing out Latitude and Longitude in degrees and decimal minutes North and West.

Subroutine for determination of 50km square reference

Optional routine for printing out 50km square letters and number code. This divides the 100km square exactly at 50km and may not give the standard result for squares adjoining a 6° meridian.

Subroutine for input of Latitude and Longitude

Input: N\$ = Latitude (degrees North).
 MINN\$ = Latitude (minutes).
 EW\$ = East or West Longitude. (Default West).
 W\$ = longitude (degrees from Greenwich).
 MINW\$ = Longitude (minutes).

Output: L(4) = Longitude East of Greenwich (degrees).
 L(5) = Latitude North (degrees).

Test Data

Irish reference	Lat./Long.	U.T.M. reference	
C000000	54°50.86'N 008°00.00'W	29 U NA	642784 NA.3
D280188	55°00.01'N 005°59.97'W	30 U UF	081990 UF.1
D279188	55°00.01'N 006°00.07'W	29 U PA	918990 PA.3
B361225	55°02.74'N 008°59.98'W	29 U NB	000000 NB.2
N000500	53°30.00'N 008°00.00'W	29 U NV	663284 NV.4
C744376	55°10.79'N 006°49.92'W	29 U PB	380170 PB.2

ACKNOWLEDGEMENTS

Dr Marc Van der Herrewegen and Jean Pierre Mouton (Belgian National Geographic Institute, Brussels) helped with geodesic matters. The junior authors would like to make it clear that the original version of the programme, and all the mathematics that this involves, would not be available except for the time and expertise in programming that Pierre Rasmont put into its production. The original version in Sharp Basic was converted to Microsoft Basic by Peter Dodd and Ray Keogh of the Forest and Wildlife service, with help in debugging by Des Higgins. Martin Bogues of the Ulster Museum helped in modifying the screen presentation and in validation of the input and making the programme more "user-friendly". Finally Eddie Wymer and Tom Curtis made helpful comments on the first draft of these accompanying notes.

REFERENCES

- Speight, M.C.D. (1977) What is a biological record: *Bull. Ir. biogeog. Soc.* 1:48-50.
 ——— & Chandler, P. J. (1983) Irish Otitidae and Platystomatidae (Diptera) including a key to the genera known in Ireland and/or Great Britain. *Ir. Nat. J.* 21:130-136.
 ———, Irwin, A. G. & de Courcy Williams, M. (1985) A provisional list of Irish Psilidae (Diptera). *Bull. Ir. biogeog. Soc.* 8:2-12.

20

01-1

IRISH NATURALISTS JOURNAL

Reprinted from

Vol. 22 No. 2 (p. 45-84)

April 1986